Имеется 1С:Предприятие 8.3 (8.3.21.1393)
При обновлении Бухгалтерия для Беларуси, редакция 2.1. с версии 2.1.29.9 на 2.1.44.5 естественно слетели дополнительные печатные формы.
В описании обновления было ЭТО и мне кажется, что это и есть причинаОбновлена «1С:Библиотека стандартных подсистем» до версии 3.1.2.527.
Обновлена «1С:Библиотека технологии сервиса» до версии 2.0.1.35.

Сравнение новой/старой версии делал и вручную переносил необходимые строки.
Но вылазит ошибка:Метод объекта не обнаружен (ПолучитьМакет)
{Документ.СчетНаОплатуПокупателю.МодульМенеджера(430)}:Макет = УправлениеПечатью.ПолучитьМакет("Документ.СчетНаОплатуПокупателю.ДоговорРБ");
{Документ.СчетНаОплатуПокупателю.МодульМенеджера(120)}:УправлениеПечатью.ВывестиТабличныйДокументВКоллекцию(КоллекцияПечатныхФорм, "ДоговорРБ", "Договор РБ",
{ОбщийМодуль.УправлениеПечатью.Модуль(2020)}:МенеджерПечати.Печать(ОбъектыСоответствующиеПечатнойФорме, ПараметрыПечати, ВременнаяКоллекцияДляОднойПечатнойФормы,
{ОбщаяФорма.ПечатьДокументов.Форма(503)}:ПечатныеФормы = УправлениеПечатью.СформироватьПечатныеФормы(Параметры.ИмяМенеджераПечати, ИменаМакетов,
{ОбщаяФорма.ПечатьДокументов.Форма(43)}:КоллекцияПечатныхФорм = СформироватьПечатныеФормы(Параметры.ИменаМакетов, Отказ);
{ОбщийМодуль.УправлениеПечатьюКлиент.Модуль(34)}:ОткрытьФорму("ОбщаяФорма.ПечатьДокументов", ПараметрыОткрытия, ВладелецФормы, Строка(Новый УникальныйИдентификатор));
{ОбщийМодуль.УправлениеПечатьюБПКлиент.Модуль(16)}:УправлениеПечатьюКлиент.ВыполнитьКомандуПечати(ОписаниеКоманды.МенеджерПечати, ОписаниеКоманды.Идентификатор, ОписаниеКоманды.ОбъектыПечати,
{(1)}:УправлениеПечатьюБПКлиент.ВыполнитьКомандуПечати(ОписаниеКоманды)
{ОбщийМодуль.УправлениеПечатьюСлужебныйКлиент.Модуль(113)}:Результат = Вычислить(Обработчик);
{ОбщийМодуль.УправлениеПечатьюСлужебныйКлиент.Модуль(26)}:ВыполнитьПодключаемуюКомандуПечатиЗавершение(Истина, ПараметрыВыполнения);
{ОбщийМодуль.ПодключаемыеКомандыКлиент.Модуль(214)}:ВыполнитьОбработкуОповещения(Обработчик, ПараметрКоманды);
{ОбщийМодуль.ПодключаемыеКомандыКлиент.Модуль(60)}:ПродолжитьВыполнениеКоманды(Контекст);
{Документ.СчетНаОплатуПокупателю.Форма.ФормаСписка.Форма(310)}:ПодключаемыеКомандыКлиент.ВыполнитьКоманду(ЭтотОбъект, Команда, Элементы.Список);

А вот собственно этадоработка:

```1c
    Функция ПечатьДоговораРБ(МассивОбъектов, ОбъектыПечати, СПечатью = Ложь)

    Перем ПодразделениеОтветственныхЛиц;

    ТабличныйДокумент = Новый ТабличныйДокумент;
    ТабличныйДокумент.АвтоМасштаб = Истина;
    ТабличныйДокумент.ОриентацияСтраницы = ОриентацияСтраницы.Портрет;
    ТабличныйДокумент.ИмяПараметровПечати = "ПАРАМЕТРЫ_ПЕЧАТИ_ЗаказПокупателя_ДоговорРБ";

    ДополнительнаяКолонкаПечатныхФормДокументов = Константы.ДополнительнаяКолонкаПечатныхФормДокументов.Получить();
    Если НЕ ЗначениеЗаполнено(ДополнительнаяКолонкаПечатныхФормДокументов) Тогда
    ДополнительнаяКолонкаПечатныхФормДокументов = Перечисления.ДополнительнаяКолонкаПечатныхФормДокументов.НеВыводить;
    КонецЕсли;
    ВыводитьКоды = ДополнительнаяКолонкаПечатныхФормДокументов <> Перечисления.ДополнительнаяКолонкаПечатныхФормДокументов.НеВыводить;

    Запрос = Новый Запрос;
    Запрос.УстановитьПараметр("МассивОбъектов", МассивОбъектов);
    Запрос.УстановитьПараметр("ДополнительнаяКолонкаПечатныхФормДокументов", ДополнительнаяКолонкаПечатныхФормДокументов);
    Запрос.Текст = ПолучитьТекстЗапросаДляФормированияПечатнойФормыСчетаНаОплат­уПокупателю();

    Шапка = Запрос.Выполнить().Выбрать();

    ПервыйДокумент = Истина;

    Пока Шапка.СледующийПоЗначениюПоля("Ссылка") Цикл

    Макет = УправлениеПечатью.ПолучитьМакет("Документ.СчетНаОплатуПокупателю.ДоговорРБ");

    ОбластьМакета = Макет.ПолучитьОбласть("Заголовок");
    ОбластьМакета.Параметры.Заполнить(Шапка);
    ОбластьМакета.Параметры.Номер = Шапка.ДоговорКонтрагента.Номер; // ПрефиксацияОбъектовКлиентСервер.ПолучитьНомерНаПечать(Шапка.Номер, Истина, Истина);;
    ОбластьМакета.Параметры.Дата = Формат(Шапка.ДоговорКонтрагента.Дата,"ДЛФ=ДД"); //Формат(Шапка.Дата,"ДЛФ=ДД");
    СведенияОКонтрагенте = БухгалтерскийУчетПереопределяемый.СведенияОЮрФизЛице(Шапка.Покупатель, Шапка.Дата);
    ОбластьМакета.Параметры.Получатель = ОбщегоНазначенияБПВызовСервера.ОписаниеОрганизации(СведенияОКонтрагенте,"НаименованиеДляПечатныхФорм" ,);

    Если ЗначениеЗаполнено(Шапка.Покупатель.ОсновноеКонтактноеЛицо) Тогда
    ОбластьМакета.Параметры.ФИО = ОбщегоНазначенияБПВызовСервера.ПолучитьФамилиюИмяОтчество(Шапка.Покупатель.ОсновноеКонтактноеЛицо.Фамилия,Шапка.Покупатель.ОсновноеКонтактноеЛицо.Имя,Шапка.Покупатель.ОсновноеКонтактноеЛицо.Отчество,,Истина);
    Иначе
    ОбластьМакета.Параметры.ФИО = "______________";

    КонецЕсли;
    ТабличныйДокумент.Вывести(ОбластьМакета);

    ОбластьМакетаЗаголовок2 = Макет.ПолучитьОбласть("Заголовок2");
    ТабличныйДокумент.Вывести(ОбластьМакетаЗаголовок2);

    ОбластьМакетаСтрока = Макет.ПолучитьОбласть("Строка");

    НомерСтроки=0;

    ИтогоСумма =0;
    ИтогоСуммаНДС = 0;
    ИтогоВсего = 0;

    Пока Шапка.Следующий() Цикл

    Если Шапка.Номенклатура = Null Тогда
    Продолжить;
    КонецЕсли;

    НомерСтроки = НомерСтроки + 1;

    ОбластьМакетаСтрока.Параметры.НомерСтроки = НомерСтроки;

    ОбластьМакетаСтрока.Параметры.Заполнить(Шапка);
    ОбластьМакетаСтрока.Параметры.Номенклатура = Шапка.Номенклатура.НаименованиеПолное;
    Всего = Шапка.Сумма + Шапка.СуммаНДС;
    ОбластьМакетаСтрока.Параметры.СуммаНДС = Формат(Шапка.СуммаНДС,"ЧДЦ=2") ;

    ОбластьМакетаСтрока.Параметры.Всего = Формат(Всего,"ЧДЦ=2") ;
    ТабличныйДокумент.Вывести(ОбластьМакетаСтрока);

    ИтогоСумма = ИтогоСумма + Шапка.Сумма;
    ИтогоСуммаНДС = ИтогоСуммаНДС + Шапка.СуммаНДС;
    ИтогоВсего = ИтогоВсего + Всего;

    Всего=0;
    КонецЦикла;

    ОбластьМакетаИтоги = Макет.ПолучитьОбласть("Итоги");

    ОбластьМакетаИтоги.Параметры.ИтогоСумма = Формат(ИтогоСумма,"ЧДЦ=2");
    ОбластьМакетаИтоги.Параметры.ИтогоСуммаНДС = Формат(ИтогоСуммаНДС,"ЧДЦ=2");
    ОбластьМакетаИтоги.Параметры.ИтогоВсего = Формат(ИтогоВсего,"ЧДЦ=2");

    ВсегоПроп = ОбщегоНазначенияБПВызовСервера.СформироватьСуммуПрописью(ИтогоВсего,Шапка.ВалютаДокумента, );
    ОбластьМакетаИтоги.Параметры.ВсегоПроп = ВсегоПроп;

    СуммаНДСПроп = ОбщегоНазначенияБПВызовСервера.СформироватьСуммуПрописью(ИтогоСуммаНДС,Шапка.ВалютаДокумента, );
    ОбластьМакетаИтоги.Параметры.СуммаНДСПроп = СуммаНДСПроп;

    ТабличныйДокумент.Вывести(ОбластьМакетаИтоги);

    ОбластьМакетаУсловия = Макет.ПолучитьОбласть("Условия");

    ДатаОплаты = Шапка.Дата + 15*24*3600;

    ОбластьМакетаУсловия.Параметры.ДатаОплаты = Формат(ДатаОплаты,"ДЛФ=Д");

    Запрос = Новый Запрос;
    Запрос.УстановитьПараметр("Клиент", СведенияОКонтрагенте.ПолноеНаименование); // Шапка.Получатель.Наименование);
    Запрос.Текст =
    "ВЫБРАТЬ
    | Контрагенты.БИН,
    | КонтрагентыКонтактнаяИнформация.Город КАК Город,
    | КонтрагентыКонтактнаяИнформация.АдресЭП КАК Адрес,
    | Контрагенты.НаименованиеПолное
    |ИЗ
    | Справочник.Контрагенты КАК Контрагенты
    | ЛЕВОЕ СОЕДИНЕНИЕ Справочник.Контрагенты.КонтактнаяИнформация КАК КонтрагентыКонтактнаяИнформация
    | ПО Контрагенты.Ссылка = КонтрагентыКонтактнаяИнформация.Ссылка
    |ГДЕ
    | Контрагенты.НаименованиеПолное = &Клиент";

    Выборка = Запрос.Выполнить().Выбрать();

    Пока Выборка.Следующий() Цикл
    Если выборка.город = " Тогда
    ОбластьМакетаУсловия.Параметры.ГородПоставки = "_____________"
    Иначе
    ОбластьМакетаУсловия.Параметры.ГородПоставки = Выборка.Город;
    КонецЕсли;

    КонецЦикла;

    ТабличныйДокумент.Вывести(ОбластьМакетаУсловия);

    ОбластьМакетаПодвал = Макет.ПолучитьОбласть("Подвал");
    ОбластьМакетаПодвал.Параметры.Получатель = СведенияОКонтрагенте.НаименованиеДляПечатныхФорм;
    ОбластьМакетаПодвал.Параметры.Адрес = СведенияОКонтрагенте.ЮридическийАдрес;
    ОбластьМакетаПодвал.Параметры.УНП = СведенияОКонтрагенте.ИНН;
    ОбластьМакетаПодвал.Параметры.ОКПО = СведенияОКонтрагенте.КодПоОКПО;
    ОбластьМакетаПодвал.Параметры.рс = СведенияОКонтрагенте.НомерСчета;
    ОбластьМакетаПодвал.Параметры.БИК = СведенияОКонтрагенте.БИК;

    Если ЗначениеЗаполнено(Шапка.Покупатель.ОсновноеКонтактноеЛицо) Тогда
    ОбластьМакета.Параметры.ФИО = ОбщегоНазначенияБПВызовСервера.ПолучитьФамилиюИмяОтчество(Шапка.Покупатель.ОсновноеКонтактноеЛицо.Фамилия,Шапка.Покупатель.ОсновноеКонтактноеЛицо.Имя,Шапка.Покупатель.ОсновноеКонтактноеЛицо.Отчество,,Истина);
    Иначе
    ОбластьМакета.Параметры.ФИО = "______________";
    КонецЕсли;

    ТабличныйДокумент.Вывести(ОбластьМакетаПодвал);

    ном = Шапка.ДоговорКонтрагента.Номер;//ПрефиксацияОбъектовКлиентСервер.ПолучитьНомерНаПечать(Шапка.Номер, Истина, Истина);;

    ТабличныйДокумент.НижнийКолонтитул.ТекстСправа = "ДОГОВОР № "+ ном + " стр.[&НомерСтраницы] из [&СтраницВсего]";
    ТабличныйДокумент.НижнийКолонтитул.Выводить = Истина;

    КонецЦикла;

    Возврат ТабличныйДокумент;

    КонецФункции
```
---
Попробуйте вместо

```1c
    УправлениеПечатью.ПолучитьМакет
```

написать

```1c
    УправлениеПечатью.МакетПечатнойФормы
```
---
УправлениеПечатью.ПолучитьМакет Просто в общем модуле - управление печатью нет теперь процедуры получить макет. Вам надо найти где она теперь или чем ее заменили и переназначить в вызове печатной формы.
---
Будем иметь ввиду вашу доработку при обновлении.
---
Опубликуйте ошибку, которая "вылазит".
---
Вообще-то она опубликована.
---
Попробуйте вместо

```1c
    УправлениеПечатью.ПолучитьМакет
```

написать

```1c
    УправлениеПечатью.МакетПечатнойФормы
```
---
Заработало
Низкий ВАМ поклон!